Lecture 12.A - Circuit Simulation and SPICE

Kael Hanson

November 17, 2015

1 SPICE Analog Simulation

SPICE is a computer program that numerically simulates electronic circuits. Many different types of analyses are possible:

- Time-domain (*i.e.* transient) analysis;
- DC analysis;
- Transfer function analysis;
- AC analysis;
- Noise analysis.

The SPICE codes have their origins 4 decades ago at UC Berkeley. Originally FORTRAN, SPICE was converted to C in the 1980's to become SPICE-3. It is an analog circuit simulation. Digital simulation or mixed-mode simulators are extensions to SPICE-3. XSPICE was developed at Georgia Tech and is a free mixed-mode simulator.

1.1 Where to get it

SPICE-3 and XSPICE are public domain software codes. Downloading the sources is certainly possible, however, some other options may be more interesting:

ngspice: an open source simulation suite which uses SPICE 3f5 and XSPICE to provide mixed-signal functionality. It is launched from the command-line but does contain a GUI for examining plots.

LT SPICE IV: a commercial but free application from Linear Technologies. This program has a schematic entry GUI so that you don't really need to know about `.cir` files. LT SPICE IV also allows 'cross-probing' for graphics output. The schematic graphics are IMO not aesthetically pleasant. It is 'multi platform' (these days means Windows, OSX) - no Linux unfortunately.

TINA TI: The Texas Instruments' circuit simulator which is in reality a limited version of the 3rd party TINA simulator. Schematic graphics are nicer, ironically the plots are not.

Despite LT and TI desiring to push their own devices, their simulators both allow one to import generic SPICE models and device subcircuits from anywhere. They make it easy to use their own parts by supplying rather extensive libraries containing the vendors' own devices.

1.2 How SPICE Views Circuits

Free schematic entry software tools notwithstanding, it's still instructive and empowering to know how to construct SPICE circuits, *and* necessary if you are using one of the tools which doesn't contain a schematic entry GUI. SPICE circuits are broken down into equipotential nodes and elements connecting them. For example the two-pole lowpass filter of Figure 1 has 4 distinct nodes: **A**, **B**, **C**, and the special node **0** which always must be present in any circuit and which represents the circuit ground.



Figure 1: A simple SPICE circuit.

1.3 The Input Deck

Circuits are described by ASCII files conventionally named with file extension `.cir`. The circuit topology is defined by a sequence of lines each containing a circuit element connecting two or more nodes. The first line of the file is a title, printed in the printout and can be any text. Lines beginning with '*' characters are comment lines. The last line of the file typically contains the `.end`

directive. The above circuit of Figure 1 would be written as:

```
A two-pole RC filter
* Lines beginning with a '*' are comments
R1     A B  1k
C1     B 0  50n
R2     B C  220
C2     C 0  10n
Rload C 0  1MEG
.end
```

For two-terminal passives such as resistors, capacitors, and inductors, the syntax is

```
Rx     <node-1> <node-2> <value>
```

For capacitors, the `R` is replaced by a `C`, or `L` for inductors. Values can be written as typical numbers, even scientific notation, or using the standard suffixes: p, n, u, m, k, MEG, G. Note that 10^6 is written `MEG` and not `M` since SPICE is case-insensitive and that would mean 10^{-3}! I've been bitten by this several times.

With a GUI-based simulator like LT SPICE, it may not be obvious how to enter SPICE circuits via text files instead of the usual schematic capture window. In fact it is seemingly not possible to create new text files in LT SPICE so you will need some ASCII text editor to create the files, however, strangely enough, once you have a text file, LT SPICE has a decent editor with colorized syntax highlighting. First create a file called `something.cir` using whatever means then open that file in LT SPICE.

1.4 Sources and Analyses

This is a valid if not interesting SPICE file. Presumably you want to *do* something with it. First, there is no stimulus. Let's add a voltage source:

```
Vsrc A 0  DC 0 AC 1
```

Finally, SPICE needs a command to tell it what information you are looking for from it. SPICE commands begin with the period. We want to do an AC analysis on this circuit so we will add the `.AC` analysis directive. This will scan a source flagged with the `AC` parameter over the specified range. The `oct` keyword indicates that the sweep is in octaves with 10, the following argument, specifying how many points per octave. Other sweep types are `dec`, for decade sweeps, and `lin` for linear sweeps:

```
.ac oct 10 1000 100Meg
```

Here is the complete `.cir` file. Try running it yourself. If you are using LT SPICE IV you can open a netlist file from the **File > Open** menu dialog.

```
A Simple SPICE Circuit
* Lines beginning with a '*' are comments
R1 A B 1k
C1 B 0 50n
R2 B C 220
C2 C 0 10n
Rload C 0  1E6
Vsrc A 0 DC 0 AC 1
.ac oct 10 1000 100Meg
.end
```

1.5 Active Devices

The real power of SPICE is its ability to simulate nonlinear complex circuits. Let's look at a simple transistor amplifier model:



Figure 2: A more involved SPICE circuit: the common-emitter amplifier.

And, it's representation in SPICE. Note the `.model` directive which defines the model for a bipolar junction transistor (selected by the `Q` circuit element). There are many parameters of the *Gummel-Poon* model. See the documentation which accompanies each SPICE for a full list of supported parameters.

```
BJT CE amplifier
V10V     Vcc 0   10
R1       Vcc B   10k
R2       B   0   2.2k
Cin      Nin B   2.2u
Q1       C B E   GenericBJT
```

```
Rc      Vcc C   10k
Re      E   0   3.3k
Rload   Nout 0    1Meg
Cout    C    Nout 2.2u
Vsrc    Nin  0    Pulse(0 1 10u 10n 10n 5u 100u)
.tran   50u
.model  GenericBJT NPN(
+ Bf=300,Vaf=100,
+ Is=5f,Cjc=4p,
+ Cje=10p,Rc=0.1,Re=0.1)
.end
```

Note that the `.tran` directive was used here to effect a time-domain simulation last for $50\,\mu$s.

1.6 Subcircuits

Actually the real power of SPICE is circuit re-use. A very common use case is this: you are designing a circuit and want to test the response of a commercially available IC. Let's take the example of an amplifier circuit which uses TI's TL972 JFET-input op-amp. TI makes their models available for download. Note that I am using a TI part with LT SPICE; SPICE is practically vendor-neutral. Here's the circuit:



Figure 3: A SPICE circuit which (re-)uses a vendor model distributed as a subcircuit in a library file.

Using pre-packaged libraries requires you to tell SPICE that you are referencing a sub-circuit or other element from an external library. This is where system dependencies get a bit hairy: how do you specify the full pathname, case-sensitivity, etc. You'll have to read the fine print here. Whatever may be the system dependencies, the library is declared with the `.lib` directive. The subcircuit is *called* or instantiated (get used to this word) using the `X` element.

```
Op-amp amplifier circuit using TI's TL972
.lib    TL972.lib
V12P    Vcc 0    12
V12N    Vee 0    -12
* This next line 'instantiates' the TL972
XU1     In+ In-  Vcc Vee Vout TL972
Rfb1    Vout In- 10k
Rfb2    In- 0    1k
Rload   Vout 0   1Meg
Vsrc    In+  0   Pulse(0 0.1 10u 10n 10n 5u 100u)
.tran   50u
.end
```

2 Mixed-Signal Simulation

LT SPICE does include limited support for mixed signal simulation. The documentation is pretty patchy and only a dozen-ish objects support this mode: you get AND, OR, NOT, and XOR logic gates (why didn't they make a NAND or NOR!), a D flip-flop, and Schmitt inverters and buffers. Some analogic aspects can be simulated (impedances, propagation delays, ...). Let's try to simulate the digital one-shot we developed a bit back:

$$\begin{aligned} D_0 &= \bar{Q}_0\bar{Q}_1T \\ D_1 &= (Q_0 \oplus Q_1)T \end{aligned}$$

Here is a schematic of the circuit:



Figure 4: Schematic of the digital one-shot FSM with node annotations for the LT SPICE mixed signal simulation.

The LT SPICE mixed simulation devices are circuit elements of type `A`. There are 8 connection nodes: 5 inputs, 2 outputs (the $\bar{Q}$ followed by Q) and one *device common* port, port 8, which any unused inputs and output should be tied to. Note that there are no power pins. The logic signal levels can be set by device parameter settings. I leave at the default of 0V for low and 1V for high. Note that if the risetime is not set the simulator complains about edge times.

```
Digital One-Shot Simulation
* Clocks, triggers, and reset logic
Vclk  Clk 0    Pulse(0 1 30n 0.5n 0.5n 10n 20n)
VTrig T   0    Pulse(0 1 55n 5n 5n 120n 1000u)
*Vrst RST 0    Pulse(0 1 0 5n 5n 25n 1)
RRst  RST 0    500
* The DFF registers
AR0  D0  0 Clk 0 RST Qb0 Q0 0 DFLOP Td=2n Trise=0.5n
AR1  D1  0 Clk 0 RST Qb1 Q1 0 DFLOP Td=2n Trise=0.5n
* The state logic
AU1   T A 0 0 0 0 D1 0 AND Td=2n Trise=0.5n
AU2   Q0 Q1 0 0 0 0 A 0 XOR Td=2n Trise=0.5n
AU3   T Qb0 Qb1 0 0 0 D0 0 AND Td=2n Trise=0.5n
.tran 250n
.end
```